Understanding Someone Else's ACS Routines

Neal Bohling and Chris Taylor
IBM

August 13, 2015
Session# 17836
Agenda

• The Problem
• Quick ACS Review
• Tools and Preparation
• Scenarios
  • Debugging ACS
  • Adding a Rule
• Preparing for Transitions
• Considering a Rewrite?
• General ACS Tips

```
IF &LABEL /= &DSN(3)
  THEN
  DO
    WRITE '&LABEL NOT SET AS INDICATED BY &DSN(3)'
    WRITE '&LABEL NOT VERIFIED BY DATACLAS'
  END
ELSE
  IF &DSN(4) = 'ONE'
    THEN
      IF &FILENUM /= 1
        THEN
          WRITE '&FILENUM NOT 1 AS INDICATED BY &DSN(4)'
          WRITE '&FILENUM NOT VERIFIED BY DATACLAS'
      END
    ELSE
      WRITE 'DATACLAS ACS ROUTINE - VERIFIED &FILENUM = '&FILENUM' '
    END
  ELSE
    IF &DSN(4) = 'TWO'
      THEN
        IF &FILENUM /= 2
          THEN
            WRITE '&FILENUM NOT 2 AS INDICATED BY &DSN(4)'
            WRITE '&FILENUM NOT VERIFIED BY DATACLAS'
        ELSE
          WRITE 'DATACLAS ACS ROUTINE - VERIFIED &FILENUM = '&FILENUM' '
        END
      ELSE
        IF &DSN(4) = 'TWO'
          THEN
            IF &FILENUM /= 2
              THEN
                WRITE '&FILENUM NOT 2 AS INDICATED BY &DSN(4)'
                WRITE '&FILENUM NOT VERIFIED BY DATACLAS'
            ELSE
              WRITE 'DATACLAS ACS ROUTINE - VERIFIED &FILENUM = '&FILENUM' '
            END
          ELSE
            WRITE 'DATACLAS ACS ROUTINE - VERIFIED &FILENUM = '&FILENUM' '
          END
        END
      END
    ELSE
      WRITE '&FILENUM NOT 1 AS INDICATED BY &DSN(4)'
      WRITE '&FILENUM NOT VERIFIED BY DATACLAS'
    END
  END
END
```
The Problem

- All routines require periodic updates
- Multiple writers may have different styles
- Special clauses can get added

- Over time, these updates can cloud the original intent and logic

Complete your session evaluations online at www.SHARE.org/Orlando-Eval
The Solution!

- Unfortunately there is no easy solution.
- But you can do to make it easier!

The Purpose of this Presentation:
- Demonstrate how to find the info you need
- Tips and tricks for making current ACS easier to read
- Discuss concepts for good ACS

Complete your session evaluations online at www.SHARE.org/Orlando-Eval
SMS Review

SMS has 4 classes:

- **Data class** – assigns allocation defaults (like size, volcnt)
- **Storage class** – assigns performance attributes
- **Management class** – defines backup characteristics
- **Storage group** – groups of volumes

Each class has its own ACS routine

*required to be SMS-managed
What is ACS?

- Automated Class Selection
- User-defined rules
- Assigns SMS classes

- **YOU** tell **SMS** how to act
Sample ACS

PROC STORCLAS
FILTLIST DBVOLS INCLUDE(IMS*,DB2*) /* ALL DATABASE VOLUMES */
   EXCLUDE('IMS053','DB2007')
FILTLIST DBJOBS INCLUDE(IMS*,PROD*,ACCT*) /* ALL DATABASE JOBS */
FILTLIST VALID_UNITS
   INCLUDE('3330','3340','3350','3375','3380','3390','SYSDA', '') /* VALID UNITS FOR SMS */
IF &UNIT ^= &VALID_UNITS
   THEN DO
      SET &STORCLAS = ''
      WRITE 'INVALID UNIT TYPE FOR SMS ALLOCATION'
      EXIT
   END
SELECT
   WHEN (&DSN = SYS1.**) /* SYSTEM DATA */
      SET &STORCLAS = 'SYSTEM'
   WHEN (&ALLVOL = &DBVOLS) && (&JOB = &DBJOBS) /* DATABASE DATA */
      SET &STORCLAS = 'DBPOOL'
   WHEN ((&DSN(3) = 'CLEAR') | (&ANYVOL ^= TSO*)) /* NON-SMS DATA */
      SET &STORCLAS = ''
   WHEN (&DEF_STORCLAS ^= '') /* IF DEFAULTS EXIST */
      SET &STORCLAS = &DEF_STORCLAS;
   OTHERWISE /* ALL OTHER DATA */
      SET &STORCLAS = 'COMMON'
END

Complete your session evaluations online at www.SHARE.org/Orlando-Eval
ACS Syntax Notes

- Always start with a **PROC**
- **END** everything – DO, PROC, SELECT
- **IF** needs **THEN**
- **SELECT** should have **WHEN** and **OTHERWISE**
- /* Comments look like this */
- Literals are in quotes: 'MY.DATASET' or '3390'
- Masks are not in quotes: MY.* or 33*
- +/- continue literals to the next line
Processing Order

1. **JOB Starts**
   - MGMTCLAS Routine
   - STORGRP Routine

2. **DATACLAS Routine**
   - DC

3. **STORCLAS Routine**
   - SC

4. **Volume Selection**
   - VOL

5. **Data Set Allocated**
ACS Philosophy

Several Approaches to ACS routines:

Speedy  Clever  Maintainable

(note: not mutually exclusive)
Best Method

- **No such thing** – very dependent on needs
- Good idea to lean toward **MAINTAINABLE**
- Don’t overcomplicate - **KISS**

Complete your session evaluations online at www.SHARE.org/Orlando-Eval
The difference between this ACS routine and the Matrix is that at least Neo can read the Matrix.
Part 2 – Preparation and Tools
Step 1 – Make a Pass through ACS

- Make a copy of your ACS and SCDS
  - For safety – do any edits on the copy
  - Example JCL at end of presentation
- Indent <optional>
  - Indent after every DO, IF, SELECT
  - Remove indent at every END, ELSE
  - Many text-editors have auto-indent
- Comment where you can -
  - /* document what you already know */
- Add WRITE statements to help trace routines
- Write down all the variables involved in decisions
Example

Old:

IF &DSN(1) = 'TEST' THEN DO
    IF &UNIT ^= &VALID_UNITS
        THEN DO
        SET &STORCLAS = ''
        EXIT
    END
ELSE SET &STORCLAS = 'TEST'
END

New:

/* Check for test data */
IF &DSN(1) = 'TEST' THEN DO
    IF &UNIT ^= &VALID_UNITS THEN DO
        /* Check if UNITs are VALID */
        IF &UNIT ^= &VALID_UNITS THEN DO
            /* if invalid, set NULL */
            SET &STORCLAS = ''
            WRITE 'INVALID UNIT TYPE'
            EXIT
        END
ELSE SET &STORCLAS = 'TEST'
    END
ELSE DO
    SET &STORCLAS = 'TEST'
    WRITE 'TEST DATA'
END
END
Step 2 – Chart the Logic

- Spreadsheet or Table
- Variables across the top
- Classes down the side

<table>
<thead>
<tr>
<th></th>
<th>A</th>
<th>B</th>
<th>C</th>
<th>D</th>
<th>E</th>
<th>F</th>
<th>G</th>
<th>H</th>
<th>I</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td></td>
<td>DSTYPE</td>
<td>JOB</td>
<td>DSORG</td>
<td>DSN(2)</td>
<td>DSTYPE</td>
<td>SIZE</td>
<td>USER</td>
<td>DATAclas</td>
</tr>
<tr>
<td>2</td>
<td>FLATSM</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>3</td>
<td>FLATBIG</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>4</td>
<td>LIBS</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>5</td>
<td>VSAM</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>6</td>
<td>CICSVS</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>7</td>
<td>TEMPS</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>8</td>
<td>JACKS</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>9</td>
<td>WRONGDC</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>10</td>
<td>ADMIN</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
</tbody>
</table>

Complete your session evaluations online at www.SHARE.org/Orlando-Eval
Set Up Your Tools

• **Set Up Your Text Editor!**
  • Highlight DO/END pairs (ISPF HILITE LOGIC)
  • Highlight IF/ELSE pairs (ISPF HILITE LOGIC)
  • Enable auto-indent (if available)
  • Possibly build your own highlighting (if available)

• **ISMF Option 7.4 and NaviQuest – ACS Testing**
  • Run combinations of variables to see what comes out
  • Define sets of test cases and run them together
  • Compare results before and after ACS changes
Example of HILITE LOGIC

```
000001 PROC &STORCLAS
000002 ****************************
000003 /* MY ACS ROUTINES FOR STORCLAS */
000004 ****************************
000005 FILTLIST DBVOLS INCLUDE(IMS*,DB2*)
000006          EXCLUDE(IMS053, DB2007)
000007 /* ALL DATABASE VOLUMES */
000008 FILTLIST DBJOBS INCLUDE(IMS*,PROD*,ACCT*)
000009          /* ALL DATA BASE JOBS */
000010 FILTLIST VALID_UNITS INCLUDE('3330','3340','3350',
000011           '3375','3380','3390','SYSDA', '')
000012 IF &UNIT ^= &VALID_UNITS THEN DO
000013     IF &DSN(1) = 'SYS' THEN DO
000014         SET &STORCLAS = '
000015         WRITE 'INVALID UNIT TYPE FOR SMS ALLOCATION'
000016         EXIT
000017     END
000018     ELSE DO
000019         SET &STORCLAS = '
000020         EXIT
000021     END
000022 END
```
Panel Utilities Help

ACS APPLICATION SELECTION

Command ==> 

Select one of the following options:

1. Edit - Edit ACS Routine source code
2. Translate - Translate ACS Routines to ACS Object Form
3. Validate - Validate ACS Routines Against Storage Constructs
4. Test - Define/Alter Test Cases and Test ACS Routines
5. Display - Display ACS Object Information
6. Delete - Delete an ACS Object from a Source Control Data Set

If Display Option is Chosen, Specify:

CDS Name . . 'NEAL.SMS.SCDS'

(1 to 44 Character Data Set Name or 'Active')

Use ENTER to Perform Selection;
Use HELP Command for Help; Use END Command to Exit.
ACS Testing – Define Testcase

Panel Utilities Help

ACS TEST SELECTION

Command ==> 

Select one of the following Options:

1. DEFINE - Define an ACS Test Case
2. ALTER - Alter an ACS Test Case
3. TEST - Test ACS Routines

If DEFINE or ALTER Option is Chosen, Specify:

ACS Test Library . . 'NEAL.SMS.ACS'
ACS Test Member . . TSTDA___
ACS Testing – Define Testcase

ACS Test Library : NEAL.SMS.ACS
ACS Test Member : TSTDA

To ALTER ACS Test Case, Specify:
   Description ==> TESTING NEW RULE
   Expected Result DC = BDAMSTUF
   DSN (DSN/Collection Name) . . SPECIAL.JKTEST.NONVSAM
   MEMN (Object Name) . . . .
   Sysname . . . Xmode . . . . . . Def_dataclas . .
   Sysplex . . . ACSenvir . . ALLOC . . Def_mgmtclas . .
   DD . . . . . Dataclas . . . . . Def_storclas . .
   Dsorg . . . DA Mgmtclas . . . . Dsntype . . . .
   Recorg . . . Storclas . . . . . If Ext . . . .
   Dsowner . . . Size . . . . . . Space_Type . .
   Expdt . . . . Maxsize . . . . Second_Qty . .
   Retpd . . . . Blksize . . .

Use ENTER to Perform Verification; Use DOWN Command to View next Panel;
Use HELP Command for Help; Use END Command to Save and Exit; CANCEL to Exit.
ACS Testing – Running a Test

Command ===>

Select one of the following Options:

3  1. DEFINE  - Define an ACS Test Case
  2. ALTER   - Alter an ACS Test Case
  3. TEST    - Test ACS Routines

If DEFINE or ALTER Option is Chosen, Specify:

ACS Test Library . . 'NEAL.SMS.ACS'
ACS Test Member . . TSTDA
ACS Testing – Running a Test

TEST ACS ROUTINES

Command ==> 

To Perform ACS Testing, Specify:

CDS Name . . . . . . 'NEAL.SMS.SCDS' (1 to 44 Character Data Set Name or 'Active')
ACS Test Library . . 'NEAL.SMS.ACS'
ACS Test Member . . TSTDA (fully or partially specified or * for all members)
Listing Data Set . . LISTING2 (1 to 44 Character Data Set Name or Blank)

Select which ACS Routines to Test:

DC ==> Y (Y/N) SC ==> Y (Y/N) MC ==> Y (Y/N) SG ==> Y (Y/N)

Use ENTER to Perform Verification and Testing;
Use HELP Command for Help; Use END Command to Exit.
## ACS Testing - Results

### ACS Testing Results

**CDS Name**: NEAL.SMS.SCDS  
**ACS Routine Types**: DC SC MC SG  
**ACS Test Library**: NEAL.SMS.ACS

<table>
<thead>
<tr>
<th>ACS Test</th>
<th>Exit Code</th>
<th>Results</th>
</tr>
</thead>
<tbody>
<tr>
<td></td>
<td></td>
<td>---------</td>
</tr>
<tr>
<td><strong>Description</strong>: Testing New Rule</td>
<td></td>
<td></td>
</tr>
<tr>
<td><strong>Expected Result</strong>: DC = BDAMSTUF</td>
<td></td>
<td></td>
</tr>
</tbody>
</table>
| **TSTDA** | 0 | DC = BDAMSTUF  
| 0 | SC = SMS |
| **MSG**: STORCLAS = SMS | | |
| 0 | MC = NULL VALUE ASSIGNED  
| 0 | SG = SGA |
| **MSG**: STORGRP = SGA | | |
| **ACS Testing RC**: 00 | | |
NaviQuest – ISMF 11

- Subset of ISMF Panels (option 11)
- Generates Libraries of Test Cases
  - Data from ISMF Lists, DCOLLECT data, SMF data
- Compare test results from before / after changes
- Generates reports
- Can be done via BATCH
- For more info, see Share Seattle session 17045

| 6 Storage Group | - Specify Volume Names and Free Space Thresholds |
| 7 Automatic Class Selection | - Specify ACS Routines and Test Criteria |
| 8 Control Data Set | - Specify System Names and Default Criteria |
| 9 Aggregate Group | - Specify Data Set Recovery Parameters |
| 10 Library Management | - Specify Library and Drive Configurations |
| **11 Enhanced ACS Management** | - Perform Enhanced Test/Configuration Management |
| C Data Collection | - Process Data Collection Function |
| G Report Generation | - Create Storage Management Reports |
| L List | - Perform Functions Against Saved ISMF Lists |
| P Copy Pool | - Specify Pool Storage Groups for Copies |
| R Removable Media Manager | - Perform Functions Against Removable Media |
Computers are like Old Testament gods; lots of rules and no mercy.”

- Joseph Campbell
Part 3 – Scenarios
"Why was X assigned when I expected Y?"
Break it Down

Why was X assigned when I expected Y?

- Why was X assigned?
- How do we get Y assigned?
Why was X assigned?

- Two+ ways to solve this.

**Start at the Beginning**
- Trace ACS
- Take each IF that matches
- End when you find the right SET

**Pros / Cons**
- Can be long
- There could be multiple paths
- May miss fall-through logic

**Start at the End**
- Find all SETs that match
- Work backwards, noting IF/SELECT requirements
- Eliminate redundant requirements

**Pros / Cons**
- Shorter
- Need less DS information
Working Backwards - Example

- “WRONGDC” data class was incorrectly assigned
- DSNAME : 'SPECIAL.JKTEST.NONVSAM'

Step 1 – Find where “WRONGDC” is set.
- Two places, with IF just before each SET:
  - IF &DSN(2) = ‘JKTEST’
  - IF &DATAACLAS = “

Step 2 – Find IF statement before IF statements:
- IF &JOB = &CICSJOB
- ELSE (aka IF &JOB NE &CICSJOB)
PROC DATACLAS

FILTLIST ADMINS INCLUDE('BOB', 'LARRY', 'MOE')
FILTLIST CICSJOBS INCLUDE(CICS*, 'CISPECL')

/* DATA CLASSES */
/* FLATSM - flat files, small, < 50mb */
/* FLATBIG - flat files, big > 50mb */
/* LIBS - PDS, PDSE */
/* VSAM - VSAM files */
/* CICSVS - VSAM files for CICS */
/* TEMPS - temporary */
/* JACKS - jacks testing DC */
/* WRONGDC - invalid combination */

IF &DSTYPE = 'TEMP' THEN SET &DATACLAS = 'TEMPS'

IF &JOB = &CICSJOBS THEN DO
  IF &DSORG = 'VS' THEN SET &DATACLAS = 'CICSVS'
  IF &DSN(2) = 'JKTEST' THEN SET &DATACLAS = 'JACKS'
  ELSE SET &DATACLAS = 'WRONGDC'
END
ELSE DO
  IF &DSORG = 'PS' THEN DO
    IF &DSTYPE = 'TEMP' THEN SET &DATACLAS = 'TEMP'
    IF &SIZE > 50MB THEN SET &DATACLAS = 'FLATSM'
    ELSE SET &DATACLAS = 'FLATBIG'
  END
  IF &DSORG = 'PO' THEN SET &DATACLAS = 'LIBS'
  IF &USER = &ADMINS THEN DO
    SET &DATACLAS = 'ADMIN'
    EXIT
  END
  IF &DSORG = 'VS' THEN DO
    SET &DATACLAS = 'VSAM'
    IF &DSTYPE = 'TEMP' THEN SET &DATACLAS = 'TEMP'
  END
  IF &JOB = 'JKTEST' THEN SET &DATACLAS = 'JACKS'
  IF &DATACLAS = '' THEN SET &DATACLAS = 'WRONGDC'
END
IF &JOB = &CICSJOBS THEN DO
  IF &DSORG = 'VS' THEN SET &DATACLAS = 'CICSVS'
  IF &DSN(2) = 'JKTEST' THEN SET &DATACLAS = 'JACKS'
  ELSE SET &DATACLAS = 'WRONGDC'
END
ELSE DO
 ..<omitted for brevity>..

IF &DSORG = 'VS' THEN DO
  SET &DATACLAS = 'VSAM'
  IF &DSTYPE = 'TEMP' THEN SET &DATACLAS = 'TEMP'
END

IF &JOB = 'JKTEST' THEN SET &DATACLAS = 'JACKS'
IF &DATACLAS = '' THEN SET &DATACLAS = 'WRONGDC'
END
END
Working Backwards - Example

• Combine set of IF statements before SET to make a rule:
  • IF &JOB = &CICSJOB AND &DSN(2) /= 'JKTEST'
  • IF &JOB /= &CICSJOB AND &DATACLASS = " (not set)

• Now we have the 2 cases where WRONGDC gets set
• Update the table:

<table>
<thead>
<tr>
<th>&amp;JOB</th>
<th>&amp;DSN(2)</th>
<th>&amp;USER</th>
<th>&amp;DATACLASS</th>
</tr>
</thead>
<tbody>
<tr>
<td>WRONGDC</td>
<td>&amp;CICSJOBS</td>
<td>NOT 'JKTEST'</td>
<td></td>
</tr>
<tr>
<td>WRONGDC</td>
<td>NOT &amp;CICSJOBS</td>
<td></td>
<td>null</td>
</tr>
</tbody>
</table>

Complete your session evaluations online at www.SHARE.org/Orlando-Eval
Break it Down

Why was X assigned when I expected Y?

Why was X assigned?

How do we get Y assigned?
How do we get Y assigned?

• Let's say the DS is supposed to have DC = 'JACKS'

• Identify all places JACKS is set:
  • IF &DSN(2) = 'JKTEST'
  • IF &JOB = 'JKTEST'

• Find Second-level tests:
  • IF &JOB = &CICSJOBS
  • IF &JOB /= &CICSJOBS
IF &JOB = &CICSJOBS THEN DO
  IF &DSORG = 'VS' THEN SET &DATACLAS = 'CICSVS'
  IF &DSN(2) = 'JKTEST' THEN SET &DATACLAS = 'JACKS'
  ELSE SET &DATACLAS = 'WRONGDC'
END
ELSE DO
  ..<omitted for brevity>..

  IF &USER = &ADMINS THEN DO
    SET &DATACLAS = 'ADMIN'
    EXIT
  END
  ..<omitted for brevity>..

  IF &JOB = 'JKTEST' THEN SET &DATACLAS = 'JACKS'
  IF &DATACLAS = '' THEN SET &DATACLAS = 'WRONGDC'
END
END
Update Table Again

• Now we know how to get what we want:
  • &JOB = &CICSJOB AND &DSN(2) = 'JKTEST'
  • &JOB /= &CICSJOB AND &JOB = 'JKTEST'

<table>
<thead>
<tr>
<th></th>
<th>&amp;JOB</th>
<th>&amp;DSN(2)</th>
<th>&amp;USER</th>
<th>&amp;DATAACLAS</th>
</tr>
</thead>
<tbody>
<tr>
<td>WRONGDC</td>
<td>&amp;CICSJOBS</td>
<td>NOT 'JKTEST'</td>
<td></td>
<td></td>
</tr>
<tr>
<td>WRONGDC</td>
<td>NOT &amp;CICSJOBS</td>
<td></td>
<td></td>
<td>null</td>
</tr>
<tr>
<td>JACKS</td>
<td>&amp;CICSJOBS</td>
<td>JKTEST</td>
<td></td>
<td></td>
</tr>
<tr>
<td>JACKS</td>
<td>NOT &amp;CICSJOB &amp; JKTEST</td>
<td></td>
<td></td>
<td></td>
</tr>
</tbody>
</table>
So what went wrong?

- **Sort Rules by &JOB** to consolidate
  - IF &JOB = &CICSJOB
    - IF &DSN(2) = 'JKTEST' – what we want
    - ELSE – what we don't want
  - IF &JOB /= &CICSJOB
    - IF &JOB = 'JKTEST' – what we want
    - IF &DATACLAS = " - what we don't want

- We know &JOB was NOT &CICSJOB

<table>
<thead>
<tr>
<th>&amp;JOB ▼</th>
<th>&amp;DSN(2)</th>
<th>&amp;USER</th>
<th>&amp;DATACLAS</th>
</tr>
</thead>
<tbody>
<tr>
<td>WRONGDC</td>
<td>NOT &amp;CICSJOBS</td>
<td>null</td>
<td></td>
</tr>
<tr>
<td>JACKS</td>
<td>JKTEST</td>
<td></td>
<td></td>
</tr>
<tr>
<td>WRONGDC</td>
<td>&amp;CICSJOBS</td>
<td>NOT 'JKTEST'</td>
<td></td>
</tr>
<tr>
<td>JACKS</td>
<td>&amp;CICSJOBS</td>
<td>JKTEST</td>
<td></td>
</tr>
</tbody>
</table>
What went wrong?

• Only two rules left:

<table>
<thead>
<tr>
<th></th>
<th>&amp;JOB ▼</th>
<th>&amp;DSN(2)</th>
<th>&amp;USER</th>
<th>&amp;DATAACLAS</th>
</tr>
</thead>
<tbody>
<tr>
<td>WRONGDC</td>
<td>NOT &amp;CICSJOBS</td>
<td></td>
<td></td>
<td>null</td>
</tr>
<tr>
<td>JACKS</td>
<td>JKTEST</td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>WRONGDC</td>
<td>&amp;CICSJOBS</td>
<td>NOT ‘JKTEST’</td>
<td></td>
<td></td>
</tr>
<tr>
<td>JACKS</td>
<td>&amp;CICSJOBS</td>
<td>JKTEST</td>
<td></td>
<td></td>
</tr>
</tbody>
</table>

• Difference is &JOB and &DATAACLAS test…
• **However**, note that for &DATAACLAS to be null (rule 1), these other rules could not have hit:
  – IF &DSORG = ‘PS’ or ‘PO’ or ‘VS’
  – IF &USER = &ADMIN
  – IF &JOB = ‘JKTEST’
IF &JOB = &CICSJOBS THEN DO
  IF &DSORG = 'VS' THEN SET &DATAACLAS = 'CICSVS'
  IF &DSN(2) = 'JKTEST' THEN SET &DATAACLAS = 'JACKS'
  ELSE SET &DATAACLAS = 'WRONGDC'
END
ELSE DO
  ..<omitted for brevity>..
  IF &USER = &ADMINS THEN DO
    SET &DATAACLAS = 'ADMIN'
    EXIT
  END
  ..<omitted for brevity>..

IF &JOB = 'JKTEST' THEN SET &DATAACLAS = 'JACKS'
IF &DATAACLAS = '' THEN SET &DATAACLAS = 'WRONGDC'
END
END
### Finish the Table

- Update the Table with this new Info

<table>
<thead>
<tr>
<th></th>
<th>&amp;JOB ▼</th>
<th>&amp;DSN(2)</th>
<th>&amp;USER</th>
<th>&amp;DC</th>
<th>&amp;DSORG</th>
</tr>
</thead>
<tbody>
<tr>
<td>WRONGDC</td>
<td>NOT &amp;CICSJOBS</td>
<td>JKTEST</td>
<td>NOT &amp;ADMINS</td>
<td>NOT ‘PS’ or ‘PO’ or ‘VS’</td>
<td></td>
</tr>
<tr>
<td>JACKS</td>
<td>JKTEST</td>
<td></td>
<td>NOT &amp;ADMINS</td>
<td></td>
<td></td>
</tr>
<tr>
<td>WRONGDC</td>
<td>&amp;CICSJOBS</td>
<td>NOT ‘JKTEST’</td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>JACKS</td>
<td>&amp;CICSJOBS</td>
<td>JKTEST</td>
<td></td>
<td></td>
<td></td>
</tr>
</tbody>
</table>
Getting it Right

• We Now Know:
  – Variable combinations for WRONGDC
  – Variable combinations for JACKS

• So to get JACKS, what needs to change?

• We're still missing some information…
• Gather available information about the job that ran:
  • &JOB = 'MYTEST'
  • &DSN(2) = 'JKTEST'
  • &USER = 'JACK'
  • &DATACLAS = TBD
  • &DSORG = 'DA' (BDAM)
Add allocation knowns

- Add a row to the table with known information:

<table>
<thead>
<tr>
<th></th>
<th>&amp;JOB ▼</th>
<th>&amp;DSN(2)</th>
<th>&amp;USER</th>
<th>&amp;DC</th>
<th>&amp;DSORG</th>
</tr>
</thead>
<tbody>
<tr>
<td>WRONGDC</td>
<td>NOT &amp;CICSJOBS NOT 'JKTEST'</td>
<td></td>
<td>NOT &amp;ADMINs</td>
<td></td>
<td>NOT 'PS' or 'PO' or 'VS'</td>
</tr>
<tr>
<td>JACKS</td>
<td>JKTEST</td>
<td></td>
<td>NOT &amp;ADMINs</td>
<td></td>
<td></td>
</tr>
<tr>
<td>MYDS</td>
<td>'MYTEST'</td>
<td>'JKTEST'</td>
<td>'JACK'</td>
<td>?</td>
<td>'DA'</td>
</tr>
</tbody>
</table>

- FILTLIST reference:
  - FILTLIST ADMINS INCLUDE('BOB','LARRY','MOE')
  - FILTLIST CICSJOBS INCLUDE(CICS*,'CISPECLE')

Complete your session evaluations online at www.SHARE.org/Orlando-Eval
What went wrong?

- Jack was doing some CICS testing under a different job name than usual: ‘MYTEST’ instead of ‘JKTEST’

- To get DC = 'JACKS', for NON-CICS, the JOBNAME needs to be 'JKTEST'

- This was just a simple example – working through the logic this way can help with bigger, more complex ACS

- Use ACS testing to verify results!
ACS Testing Example

ACS TEST CASE ALTER

Command ==> ACS Test Library : NEAL.SMS.ACS
ACS Test Member . : TEST1

To ALTER ACS Test Case, Specify:
  Description ==> TESTING JKTEST
  Expected Result
  DSN (DSN/Collection Name) . . SPECIAL.JKTEST.NONVSAM
  MEMN (Object Name) . . .
  Sysname . .
  Sysplex . .
  ACSenvir . .
  DD . .
  Dataclas . .
  Dsorg . .
  Mgmtclas . .
  Recorg . .
  Storclas . .
  Job . .

Run once with MYTEST (TEST1)
Run again with JKTEST (TEST2)
# ACS Testing Results

ACS TESTING RESULTS

CDS NAME : NEAL.SMS.SCDS
ACS ROUTINE TYPES: DC
ACS TEST LIBRARY : NEAL.SMS.ACS

<table>
<thead>
<tr>
<th>ACS TEST MEMBER</th>
<th>EXIT CODE</th>
<th>RESULTS</th>
</tr>
</thead>
<tbody>
<tr>
<td>DESCRIPTION: TESTING JKTEST</td>
<td></td>
<td></td>
</tr>
<tr>
<td>EXPECTED RESULT:</td>
<td></td>
<td></td>
</tr>
<tr>
<td>TEST1</td>
<td>0</td>
<td>DC = WRONGDC</td>
</tr>
</tbody>
</table>

ACS TESTING RC: 00

Complete your session evaluations online at www.SHARE.org/Orlando-Eval
A few notes:

• Logic Chart works best where all SETs are mutually exclusive.
• If you have fall-through logic, you may consider using an ORDER field as well to sequence logic.

• Try to avoid testing &DATACLAS in DC routines
  – Often a failsafe, but complicates logic
  – Or remove / add WRITE when debugging.

• Adding EXIT after each SET is good practice, but can complicate logic.

• Strategic WRITE statements can help demonstrate decision logic.

• Use ACS testing to run your modified ACS to track logic.
Summary of Scenario 1:

- Tracked backwards from the SET
- Created a TABLE or MAP to detail conditions / rules
- Compared actual rules to expected result
- Found culprit
- Tested changes with ACS Tester
"To err is human - and to blame it on a computer is even more so."

- Robert Orben
Scenario 2

“We added a class, and now we need to update the routines.”
Updating Routines - Overview

- Varying difficulty, depending on new rules
- Simpler with a table or MAP

Basic steps:
- Copy and work on copy!
- Find logic section that matches
- Insert new rule
- Test, fix, test, fix
Updating Routines – Don't

- Don't just add a simple rule to the beginning:
  - IF &DSORG="VS" THEN SET &DATAACLAS = “VSAM”
  - You’ll bypass all your old routines!

- Don't just add a simple rule to the end:
  - It might never get run
  - It might overwrite a different set (watch out for fall through)

- Don't forget to comment
- Don't use variables that are not used anywhere else
  - You'll end up with non-exclusive paths
  - Try to use consistent logic
Understand the Logic

• How do the routines select classes?
• What variables are used?
• Build a table or MAP:

![Table Image]

Complete your session evaluations online at www.SHARE.org/Orlando-Eval
Adding Rules - Example

- We want to add a new class **BDAMSTUF**
  - Non-CICS
  - DSORG='DA'
- Add to the MAP:

<table>
<thead>
<tr>
<th></th>
<th>A</th>
<th>B</th>
<th>C</th>
<th>D</th>
<th>E</th>
<th>F</th>
<th>G</th>
<th>H</th>
<th>I</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td></td>
<td>CLASS</td>
<td>&amp;DSTYPE</td>
<td>&amp;JOB</td>
<td>&amp;DSORG</td>
<td>&amp;DSN(2)</td>
<td>&amp;STYPE</td>
<td>&amp;SIZE</td>
<td>&amp;USER</td>
</tr>
<tr>
<td>2</td>
<td></td>
<td>WRONGDC</td>
<td>NOT &amp;CICSJOBS</td>
<td>and NOT JKTEST</td>
<td>DA</td>
<td>NOT &amp;ADMIN</td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>3</td>
<td></td>
<td>FLATSM</td>
<td>NOT &amp;CICSJOBS</td>
<td>PS</td>
<td>&lt; 50M</td>
<td>NOT &amp;ADMIN</td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>4</td>
<td></td>
<td>FLATBIG</td>
<td>NOT &amp;CICSJOBS</td>
<td>PS</td>
<td>&gt;= 50M</td>
<td>NOT &amp;ADMIN</td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>5</td>
<td></td>
<td>LIBS</td>
<td>NOT &amp;CICSJOBS</td>
<td>PO</td>
<td></td>
<td>&amp;ADMIN</td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>6</td>
<td></td>
<td>VSAM</td>
<td>NOT &amp;CICSJOBS</td>
<td>VS</td>
<td></td>
<td>NOT &amp;ADMIN</td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>7</td>
<td></td>
<td>TEMPS</td>
<td>TEMP</td>
<td>NOT &amp;CICSJOBS</td>
<td>PS</td>
<td>VS</td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>8</td>
<td></td>
<td>ADMIN</td>
<td>NOT &amp;CICSJOBS</td>
<td></td>
<td></td>
<td>&amp;ADMIN</td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>9</td>
<td></td>
<td>JACKS</td>
<td>JKTEST</td>
<td></td>
<td></td>
<td>NOT &amp;ADMIN</td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>10</td>
<td></td>
<td>CICSVS</td>
<td>&amp;CICSJOB</td>
<td>VS</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>11</td>
<td></td>
<td>JACKS</td>
<td>&amp;CICSJOB</td>
<td>JKTEST</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>12</td>
<td></td>
<td>WRONGDC</td>
<td>&amp;CICSJOB</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>13</td>
<td></td>
<td>BDAMSTUF</td>
<td>NOT &amp;CICSJOBS</td>
<td>DA</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>14</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
</tbody>
</table>
Adding Rules - Example

- Compare to other rules by Variable
  - Only two in play: &JOB and &DSORG
- Fits in our NOT &CICSJOB section
- Fits next to other &DSORG tests
IF &DSTYPE = 'TEMP' THEN SET &DATACLAS = 'TEMPS'

IF &JOB = &CICSJOBS THEN DO
  IF &DSORG = 'VS' THEN SET &DATACLAS = 'CICSVS'
  IF &DSN(2) = 'JKTEST' THEN SET &DATACLAS = 'JACKS'
  ELSE SET &DATACLAS = 'WRONGDC'
END

ELSE DO
  IF &DSORG = 'PS' THEN DO
    IF &DSTYPE = 'TEMP' THEN SET &DATACLAS = 'TEMP'
    IF &SIZE > 50MB THEN SET &DATACLAS = 'FLATSM'
    ELSE SET &DATACLAS = 'FLATBIG'
  END

  IF &DSORG = 'PO' THEN SET &DATACLAS = 'LIBS'

  IF &USER = &ADMINS THEN DO
    SET &DATACLAS = 'ADMIN'
    EXIT
  END

  IF &DSORG = 'VS' THEN DO
    SET &DATACLAS = 'VSAM'
    IF &DSTYPE = 'TEMP' THEN SET &DATACLAS = 'TEMP'
  END

  IF &JOB = 'JKTEST' THEN SET &DATACLAS = 'JACKS'
  IF &DATACLAS = '' THEN SET &DATACLAS = 'WRONGDC'
END
Adding Rules - Example

• Insert the new rule:
  • IF &DSORG='DA' THEN SET &DATAACLAS = 'BDAMSTUF'

• This many IF statements should be a SELECT

• Use the best practices already in use in the routine
  • Don't change practices unless you're ready to rewrite

• Note your update in the CHANGELOG
  • If one doesn't exit, CREATE ONE!
PROC DATACLAS

FILTLIST ADMINS INCLUDE('BOB','LARRY','MOE')
FILTLIST CICSJOBS INCLUDE(CICS*,'CISPECL')

/* DATA CLASSES */
/* FLATSM  - flat files, small, < 50mb */
/* FLATBIG - flat files, big > 50mb */
/* LIBS    - PDS, PDSE */
/* VSAM    - VSAM files */
/* CICSVS  - VSAM files for CICS */
/* TEMPS   - temporary */
/* JACKS   - jacks testing DC */
/* WRONGDC - invalid combination */

/* CHANGE LOG */
/* 2013.08.13 - Added BDAMSTUF to NON-CICS group */
/* - Also switched from IF to SELECT block */
IF &DSTYPE = 'TEMP' THEN SET &DATACLAS = 'TEMPS'

IF &JOB = &CICSJOBS THEN DO
  IF &DSORG = 'VS' THEN SET &DATACLAS = 'CICSVS'
  IF &DSN(2) = 'JKTEST' THEN SET &DATACLAS = 'JACKS'
  ELSE SET &DATACLAS = 'WRONGDC'
END
ELSE DO
  SELECT ( &DSORG )
    WHEN ('PS') DO
      IF &DSTYPE = 'TEMP' THEN SET &DATACLAS = 'TEMP'
      IF &SIZE > 50MB THEN SET &DATACLAS = 'FLATSM'
      ELSE SET &DATACLAS = 'FLATBIG'
    END
    WHEN ('PO') SET &DATACLAS = 'LIBS'
    WHEN ('VS') DO
      SET &DATACLAS = 'VSAM'
      IF &DSTYPE = 'TEMP' THEN SET &DATACLAS = 'TEMP'
    END
    WHEN ('DA') SET &DATACLAS = 'BDAMSTUF'
    OTHERWISE SET &DATACLAS = 'WRONGDC'
  END /* END SELECT &DSORG */

IF &USER = &ADMINS THEN DO
  SET &DATACLAS = 'ADMIN'
  EXIT
END
IF &JOB = 'JKTEST' THEN SET &DATACLAS = 'JACKS'
IF &DATACLAS = '' THEN SET &DATACLAS = 'WRONGDC'
END
Test Before / After

- Defined testcase with:
  - DSN: SPECIAL.JKTEST.NONVSAM
  - DSORG: DA
  - ACSENVIR: ALLOC
  - JOB: MYTEST

<table>
<thead>
<tr>
<th>ACS TEST</th>
<th>MEMBER</th>
<th>EXIT CODE</th>
<th>RESULTS</th>
</tr>
</thead>
<tbody>
<tr>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
</tbody>
</table>

**BEFORE:**

<table>
<thead>
<tr>
<th>DESCRIPTION: TESTING NEW RULE</th>
</tr>
</thead>
<tbody>
<tr>
<td>EXPECTED RESULT:</td>
</tr>
<tr>
<td>TSTDA</td>
</tr>
<tr>
<td>0</td>
</tr>
<tr>
<td>DC = WRONGDC</td>
</tr>
</tbody>
</table>

**AFTER:**

<table>
<thead>
<tr>
<th>DESCRIPTION: TESTING NEW RULE</th>
</tr>
</thead>
<tbody>
<tr>
<td>EXPECTED RESULT:</td>
</tr>
<tr>
<td>TSTDA</td>
</tr>
<tr>
<td>0</td>
</tr>
<tr>
<td>DC = WRONGDC</td>
</tr>
</tbody>
</table>
Test, Fix, Test

- Use the ACS routine tester!
- Run a series of tests to verify new changes work
- Run a series of tests to ensure old rules still work
- If not, figure out why and fix
- Repeat
- Test again after activation

- You can use NaviQuest to build suites of tests
  - See Seattle session 17045 – NaviQuest: Streamlining SMS
Adding Rules - Summary

- Make a copy
- Understand the logic
- Find relevant section
- Update CHANCELOG
- Add rule
- Avoid breaking anything
- Test, test, test
Part 4 – Preparing for Transitions
Some tips and things to consider
HSM Transitions

• HSM Transitions feature will re-run ACS routines to assign to new classes (except DC)
• &ACSENVIR = ‘SPMGCLTR’
  – SPace ManaGement CLass TRansitions
HSM Transitions

SELECT (&ACSENVIR)
WHEN('SPMGCLTR') DO

/* Transition from one to another */
/* Re-use previous logic, but set phase 2 class */
IF( &DSN(2) = 'LOGDATA') THEN SET &STORGRP = …

/* Base decision off first assignment (not recommended) */
IF( &DATACLAS = 'PASS1' ) THEN SET &STORGRP = …

/* Completely new logic */
WRITE 'DONT FORGET THE WRITE STATEMENTS'
END

OTHERWISE DO

/* OLD LOGIC, UNTOUCHED */
IF …
END
END

Complete your session evaluations online at www.SHARE.org/Orlando-Eval
Part 5 – Considering a Rewrite?
Some tips and things to consider
Rewrite Process

• Make sure you understand the logic
• Have clearly defined rules BEFORE you start
• Consider how things might change in the future
• Group rules by variable in order of importance / frequency
• Code rules using nested SELECT / WHEN

• Create a suite of ACS Test cases (see NaviQuest)
• Run before / after to ensure logic works the same

• COMMENT your logic flow
Rewrite DOs

- Prefer SELECT over IF
- Try to make logic mutually exclusive (one big decision tree)
- Code rules in order of most to least specific
- Use FILTLIST names that are helpful
- Use COPYFILT to keep common FILTLISTS the same
- EXIT after a SET
- Use WRITE statements liberally
- Test repeatedly
Rewrite DON’Ts

• Use lists of IF statements (resist nesting)
• Mirror FILTLISTs between routines unless needed
• Base your logic of the DC or SC assignment
• Use huge numbers of variable
• Work on active SCDS / ACS
• Forget to Test, Test, Test!
References

- **DFSMS Implementing System-Managed Storage**
  - SC26-7407
- **DFSMSdfp Storage Administration**
  - SC26-7402
Notices & Disclaimers

Copyright © 2015 by International Business Machines Corporation.

No part of this document may be reproduced or transmitted in any form without written permission from IBM Corporation.

Product information and data has been reviewed for accuracy as of the date of initial publication. Product information and data is subject to change without notice. This document could include technical inaccuracies or typographical errors. IBM may make improvements and/or changes in the products and/or programs described herein at any time without notice.

References in this document to IBM products, programs, or services does not imply that IBM intends to make such products, programs or services available in all countries in which IBM operates or does business. Consult your local IBM representative or IBM Business Partner for information about the product and services available in your area.

Any reference to an IBM Program Product in this document is not intended to state or imply that only that program product may be used. Any functionally equivalent program, that does not infringe IBM's intellectually property rights, may be used instead. It is the user's responsibility to evaluate and verify the operation of any non-IBM product, program or service.

THE INFORMATION PROVIDED IN THIS DOCUMENT IS DISTRIBUTED "AS IS" WITHOUT ANY WARRANTY, EITHER EXPRESS OR IMPLIED. IBM EXPRESSLY DISCLAIMS ANY WARRANTIES OF MERCHANTABILITY, FITNESS FOR A PARTICULAR PURPOSE OR INFRINGEMENT. IBM shall have no responsibility to update this information. IBM products are warranted according to the terms and conditions of the agreements (e.g., IBM Customer Agreement, Statement of Limited Warranty, International Program License Agreement, etc.) under which they are provided. IBM is not responsible for the performance or interoperability of any non-IBM products discussed herein.

Complete your session evaluations online at www.SHARE.org/Orlando-Eval
Notices & Disclaimers

The performance data contained herein was obtained in a controlled, isolated environment. Actual results that may be obtained in other operating environments may vary significantly. While IBM has reviewed each item for accuracy in a specific situation, there is no guarantee that the same or similar results will be obtained elsewhere.

The responsibility for use of this information or the implementation of any of these techniques is a customer responsibility and depends on the customer’s or user’s ability to evaluate and integrate them into their operating environment. Customers or users attempting to adapt these techniques to their own environments do so at their own risk. IN NO EVENT SHALL IBM BE LIABLE FOR ANY DAMAGE ARISING FROM THE USE OF THIS INFORMATION, INCLUDING BUT NOT LIMITED TO, LOSS OF DATA, BUSINESS INTERRUPTION, LOSS OF PROFIT OR LOSS OF OPPORTUNITY.

Information concerning non-IBM products was obtained from the suppliers of those products, their published announcements or other publicly available sources. IBM has not necessarily tested those products in connection with this publication and cannot confirm the accuracy of performance, compatibility or another claims related to non-IBM products. Questions on the capabilities of non-IBM products should be addressed to the suppliers of those products.

The provision of the information contained herein is not intended to, and does not, grant any right or license under any IBM patents or copyrights. Inquiries regarding patent or copyright licenses should be made, in writing, to:

IBM Director of Licensing
IBM Corporation
North Castle Drive
Armonk, NY 10504-1785
U.S.A.

Any statements regarding IBM’s future direction and intent are subject to change or withdrawal without notice, and represent goals and objectives only.

Complete your session evaluations online at www.SHARE.org/Orlando-Eval
Trademarks

DFSMSdfp, DFSMSdss, DFSMSshm, DFSMSrmm, IBM, IMS, MVS, MVS/DFP, MVS/ESA, MVS/SP, MVS/XA, OS/390, SANergy, and SP are trademarks of International Business Machines Corporation in the United States, other countries, or both.

AIX, CICS, DB2, DFSMS/MVS, Parallel Sysplex, OS/390, S/390, Seascape, and z/OS are registered trademarks of International Business Machines Corporation in the United States, other countries, or both.

Domino, Lotus, Lotus Notes, Notes, and SmartSuite are trademarks or registered trademarks of Lotus Development Corporation. Tivoli, TME, Tivoli Enterprise are trademarks of Tivoli Systems Inc. in the United States and/or other countries.

Java and all Java-based trademarks are trademarks of Sun Microsystems, Inc. in the United States, other countries, or both. UNIX is a registered trademark in the United States and other countries licensed exclusively through The Open Group.

Other company, product, and service names may be trademarks or service marks of others.